Discrete Choice Modeling
William Greene
Stern School of Business, New York University

Lab Session 3

I. Bivariate and Multivariate Probit

This exercise uses the data file panelprobit.lpj

Some preliminaries after the file is loaded.

```
* Declare panel
xtset firm
* Some lists to be used later
global x84 imum84 fdium84 prod84
global x85 imum85 fdium85 prod85
global x86 imum86 fdium86 prod86
```

1. The Bivariate Probit Model.

In this exercise, we will first fit a bivariate probit model. The model is

$$y_1^* = x_1'\beta_1 + \varepsilon_1$$
$$y_2^* = x_2'\beta_2 + \varepsilon_2$$
$$\varepsilon_1,\varepsilon_2 \sim N_2[(0,0),(1,1,\rho)].$$

The model is fit by maximum likelihood. You can use the following commands to treat the 1984 and 1985 observations as a bivariate probit outcome: (Later, we will apply the model to two distinct decisions.)

```
* Bivariate probit
biprobit (ip84 = $x84) (ip85 = $x85) if t==1
scalar lu = e(ll)
```

Notice that if $\boldsymbol{\beta}_1 = \boldsymbol{\beta}_2$, that this becomes a two period random effects model. It is possible to trick the program into fitting the restricted bivariate probit model by fitting a two period random effects probit model, which will have one set of coefficients. This way, we can test the hypothesis of coefficient equality. Do the results change substantially when the restriction is imposed? Does the estimate of ρ change? The hypothesis of interest is H_0: $\boldsymbol{\beta}_1 = \boldsymbol{\beta}_2$. You can test this hypothesis using these models with a likelihood ratio test. Compute twice the difference in the log likelihoods. What are the degrees of freedom for the test? What is the 95% critical value from the chi squared table?

```
* Restrictions - equal coefficient vectors
biprobit (ip84 = $x84) (ip85 = $x85) if t==1
scalar lu = e(ll)
xtprobit ip imum fdium prod if t <= 2, re
scalar lr = e(ll)
display 2*(lu - lr)
```

2. Recursion

Many applications involve simultaneous equations sorts of binary choice models. The bivariate probit model is analogous to the seemingly unrelated regressions, save, of course for the discrete dependent variables.

The recursive bivariate probit model has made some recent appearances in the literature. A two period panel version might appear as follows:

$$y_{i1} = 1[\boldsymbol{\beta}_1'\mathbf{x}_{i1} + \varepsilon_{i1} > 0],$$
$$y_{i2} = 1[\boldsymbol{\beta}_2'\mathbf{x}_{i2} + \gamma_2 y_{i1} + \varepsilon_{i1} > 0]$$

One might model insurance take up with such a model. Here, we model innovation in 1985 as a function of 1985 covariates and 1984 innovation. An interesting feature of the recursive bivariate probit model is that variables in the second may have two effects on y_{i2}. A direct effect in $\mathbf{x}_{i2}$ and an indirect effect if they appear in the y_{i1} equation and effect y_{i1} which then affects y_{i2}.

```
biprobit (ip84 = $x84 invgood consgood) ///
    (ip85 = $x85 invgood consgood ip84) if t == 1
```

3. Identification and Incoherence

There is a temptation sometimes to specify a fully simultaneous model, as in

$$y_{i1} = 1[\boldsymbol{\beta}_1'\mathbf{x}_{i1} + \gamma_1 y_{i2} + \varepsilon_{i1} > 0],$$
$$y_{i2} = 1[\boldsymbol{\beta}_2'\mathbf{x}_{i2} + \gamma_2 y_{i1} + \varepsilon_{i1} > 0]$$

This model is not identified and is inestimable. In the common language used for discrete choice modeling, the model is ‘incoherent.’ That does not mean that one cannot try to fit the model. Unlike linear models, it is sometimes possible to obtain numbers for unidentified nonlinear models.

For this particular one, NLOGIT will refuse to try – the first command produces an error message. But, suppose we try to bypass this control, and program and estimate our own unidentified model.

```
* An incoherent model
biprobit (ip84 = $x84 invgood consgood ip85) ///
    (ip85 = $x85 invgood consgood ip84) if t == 1
* (This model should be stopped by the software)
```

4. Sample Selection

A sample selection model for binary outcomes would work the same as in the linear case. However, the model is fit by maximum likelihood, and there is no inverse Mills ratio ('lambda') involved in the estimation. Here, we select on a particular industry, investment goods, and fit a model for innovation.

```
* sample selection
heckprob ip84 $x84 if t == 1,select(invgood=prod84 im84)
```

One occasionally reads that the sample selection model requires for identification that there be at least one variable in the selection (probit) equation that is not in the main equation. Strictly, this is not true for the bivariate probit model – in fact, it is not even true for the linear model. The following illustrates. In fact, the estimator is improved when the model is 'identified' in this fashion, but statistically, it is not necessary.

```
* sample selection model does not require exclusions
heckprob ip84 prod84 imum84 fdium84 sp84 lsales84 ///
if t == 1,select(invgood=prod84 imum84 fdium84 sp84 lsales84)

* the linear model doesn't either.
heckman ip84 prod84 imum84 fdium84 sp84 lsales84 ///
if t == 1,select(invgood=prod84 imum84 fdium84 sp84 lsales84) two
```